//在不同时间矿物出产情况
public class OreStages {
    /*沙砾、煤炭、铁、金、青金石、红石*/
    //从石头里
    private static stone(coal_chunk as int,iron_nugget as int,gold_nugget as int,
        lapis_lazuli as int,redstone as int
    ) as void{
        <recipetype:create:crushing>.removeRecipe(<item:minecraft:gravel>);
        <recipetype:create:crushing>.addRecipe("meng_basics_ore",[
	        <item:minecraft:gravel> % 50,<item:morecharcoal:coal_chunk> % coal_chunk,<item:minecraft:iron_nugget> % iron_nugget,<item:minecraft:gold_nugget> % gold_nugget,
	        <item:minecraft:lapis_lazuli> % lapis_lazuli,<item:minecraft:redstone> % redstone
            ],<item:minecraft:stone>,500);  
    }
    //开始
    public static startStone() as void{
        stone(5,2,3,5,3);
    }
    //一段时间后
    public static spellStone() as void{
        stone(15,20,15,33,33);
    }
    //最后
    public static endStone() as void{
        stone(90,60,50,60,60);
    }

    /*各种鱼、墨囊、海草、海带、海泡菜、海晶碎片、海晶砂粒*/
    //从沙子里
    private static send(cod as int,salmon as int,tropical_fish as int,ink_sac as int,kelp as int,seagrass as int,
        sea_pickle as int,prismarine_shard as int,prismarine_crystals as int
    ) as void{
        <recipetype:create:crushing>.removeRecipe(<item:create:limesand>);//删除石灰沙的粉碎配方
        //<recipetype:create:crushing>,removeRecipe(<item:minecraft:sand>);
        <recipetype:create:crushing>.addRecipe("meng_sand_sea",[
	        <item:minecraft:cod> % cod,<item:minecraft:salmon> % salmon,<item:minecraft:tropical_fish> % tropical_fish,
	        <item:minecraft:ink_sac> % ink_sac,<item:minecraft:kelp> % kelp,<item:minecraft:seagrass> % seagrass,
	        <item:minecraft:sea_pickle> % sea_pickle,<item:minecraft:prismarine_shard> % prismarine_shard,<item:minecraft:prismarine_crystals> % prismarine_crystals
        ],<item:minecraft:sand>);
    }

    public static startSend() as void   {
        send(60,60,40,10,15,15,5,1,1);
    }

    public static spellSend() as void{
        send(80,80,60,30,35,35,10,5,5);
    }

    public static endSend() as void{
        send(95,95,80,60,70,70,30,15,15);
    }

    /*沙子、黏土球、煤炭、铁、金、青金石、红石*/
    //从沙砾里
    private static gravel(clay_ball as int,coal_chunk as int,iron_nugget as int,gold_nugget as int,
        lapis_lazuli as int,redstone as int
    ) as void{  
        <recipetype:create:crushing>.removeRecipe(<item:minecraft:sand>);//先删掉沙砾磨沙子的配方
       // <recipetype:create:crushing>.removeRecipe(<item:minecraft:gravel>);
        <recipetype:create:crushing>.addRecipe("meng_basics_gravel",[
	        <item:minecraft:sand> % 60,<item:minecraft:clay_ball> *2 % clay_ball,<item:morecharcoal:coal_chunk> * 6 % coal_chunk,<item:minecraft:iron_nugget>* 6 % iron_nugget,
	        <item:minecraft:gold_nugget> * 2 % gold_nugget,<item:minecraft:lapis_lazuli> * 3 % lapis_lazuli,<item:minecraft:redstone> * 4 % redstone
        ],<item:minecraft:gravel>,500);
    }

    public static startGravel() as void{
        gravel(11,3,4,4,3,6);
    }

    public static spellGravel() as void{
        gravel(33,30,40,40,30,60);
    }

    public static endGravel() as void{
        gravel(50,50,55,55,45,65);
    }
}

